



TSwap Protocol Audit Report

Version 1.0

github.com/lucasfhope

October 3, 2025

TSwap Protocol Audit Report

Lucas Hope

October 3, 2025

Prepared by: Lucas Hope

Table of Contents

- Table of Contents
- Protocol Summary
- Disclaimer
- Risk Classification
- Audit Details
 - Scope
 - Roles
- Executive Summary
 - Issues found
- Findings
 - High
 - Medium
 - Low
 - Informational
 - Gas

Protocol Summary

The protocol intends to be a decentralized exchange which uses a constant product formula to set the swap rate between WETH and a pool token. The pool token can be set as any ERC20 during pool creation. The DEX is provided funds by liquidity providers that receive liquidity tokens to represent their stake in the pool. Users can swap between tokens in a pool but each swap will include a 0.3% fee.

Disclaimer

Lucas Hope makes all effort to find as many vulnerabilities in the code in the given time period, but holds no responsibilities for the findings provided in this document. A security audit by the team is not an endorsement of the underlying business or product. The audit was time-boxed and the review of the code was solely on the security aspects of the Solidity implementation of the contracts.

Risk Classification

		Impact		
		High	Medium	Low
Likelihood	High	H	H/M	M
	Medium	H/M	M	M/L
	Low	M	M/L	L

We use the CodeHawks severity matrix to determine severity. See the documentation for more details.

Audit Details

The findings described in this report corresponds to the following commit hash:

```
1 f5a14bc4c1f202aa2f8bdc210cdfea7e1f276735
```

Scope

```
1 ./src/  
2 --- PoolFactory.sol  
3 --- TSwapPool.sol
```

Roles

- Liquidity Providers: Users who have liquidity deposited into the pools. Their shares are represented by the LP ERC20 tokens. They gain a 0.3% fee every time a swap is made.
- Users: Users who want to swap tokens.

Executive Summary

The audit found issues with certain function in the protocol that were not implemented as the protocol intended. The audit also found code and other instances that cause the core invariant of the protocol to break.

Issues found

Severity	Number of issues found
High	4
Medium	2
Low	2
Info	7
Gas	3
Total	18

Findings

High

[H-1] Incorrect fee calculation in `TSwapPool::getInputAmountBasedOnOutput` causes the protocol to take too many tokens from the users, resulting in fees that are too high.

Description: The `TSwapPool::getInputAmountBasedOnOutput` function is intended to calculate the amount of tokens should be deposited for a given output of tokens, which should include a 0.3% fee. Instead, the function calculates the fee at over 90%. This is because it scales the amount by 10_000 instead of 1_000.

Impact: Protocol takes more fees than expected from users.

Proof of Code: You can add the following test to your test suite in `test/unit/TSwapPool.t.test`.

```
1 function testFeesAreMiscalculatedInInputAmountBasedOnOutput() public {
2     // LP deposit liquidity
3     vm.startPrank(liquidityProvider);
4     weth.approve(address(pool), 100e18);
5     poolToken.approve(address(pool), 100e18);
6     pool.deposit(100e18, 100e18, 100e18, uint64(block.timestamp));
7     vm.stopPrank();
8
9     // expected swap amount with 0.3% fee, want to swap ?? PoolToken
    for 1 WETH
10    uint256 expectedPoolTokens = ((poolToken.balanceOf(address(pool)) *
        1 ether) * 1000) / ((weth.balanceOf(address(pool)) - 1 ether) *
        997); // correct fee ratio -> 997/1000
11    uint256 calculatedPoolTokens = pool.getInputAmountBasedOnOutput(
12        1 ether,
13        poolToken.balanceOf(address(pool)),
14        weth.balanceOf(address(pool))
15    );
16
17    console.log("expectedPoolTokens: ", expectedPoolTokens);
18    console.log("calculatedPoolTokens: ", calculatedPoolTokens);
19    assert(expectedPoolTokens != calculatedPoolTokens);
20 }
```

When you run the test, you will see that the calculated amount of tokens is 10 times larger than what it should be.

Recommended Mitigation:

```
1 function getInputAmountBasedOnOutput(
```

```
2     uint256 outputAmount,  
3     uint256 inputReserves,  
4     uint256 outputReserves  
5 )  
6     public  
7     pure  
8     revertIfZero(outputAmount)  
9     revertIfZero(outputReserves)  
10    returns (uint256 inputAmount)  
11 {  
12 -     return ((inputReserves * outputAmount) * 10000) / ((  
    outputReserves - outputAmount) * 997);  
13 +     return ((inputReserves * outputAmount) * 1000) / ((  
    outputReserves - outputAmount) * 997);  
14 }
```

[H-2] Lack of slippage protection in TSwapPool::swapExactOutput could cause users to receive way fewer tokens than expected.

Description: The `TSwapPool::swapExactOutput` function does not include any sort of slippage protection. This function is similar to swap is done in `TSwapPool::swapExactInput`, where the function specifies `minOutputAmount`. The `TSwapPool::swapExactOutput` function should specify a `maxInputAmount`.

Impact: If the market conditions change before the transaction process, the user would get a much worse swap.

Proof of Concept: 1. The price of 1 WETH right now is 1000 USDC 2. User inputs a `swapExactOutput` looking for 1 WETH 1. inputToken = USDC 2. outputToken = WETH 3. outputAmount = 1 4. deadline = anything 3. The function does not offer a max input amount 4. As the transaction is pending in the mempool, the market changes! The price moves a lot such that 1 WETH -> 3000 USDC, 3x more than the user expected. 5. The transaction completes but the user sends the protocol 10,000 USDC instead of 1000 USDC.

Proof of Code: You can add the following test to your test suite in `test/unit/TSwapPool.t.test`. You should also update `TSwapPool::getInputAmountBasedOnOutput` so that the fee ratio is 997 / 1000 instead of 997 / 10000.

```
1     function testSwapExactOutputHasNoSlippageProtection() public {  
2         // LP deposit liquidity  
3         vm.startPrank(LiquidityProvider);  
4         weth.approve(address(pool), 100e18);  
5         poolToken.approve(address(pool), 100e18);  
6         pool.deposit(100e18, 100e18, 100e18, uint64(block.timestamp));  
7         vm.stopPrank();
```

```
8
9     uint256 outputAmount = 5 ether;
10    uint256 inputReserves = weth.balanceOf(address(pool));
11    uint256 outputReserves = poolToken.balanceOf(address(pool));
12
13    // we want to get 5 WETH, how many PoolTokens do we need to
14    // give
15    uint256 expectedInputAmount = pool.getInputAmountBasedOnOutput(
16        outputAmount,
17        inputReserves,
18        outputReserves
19    );
20
21    // another user does a swap before to change the pool state
22    address user2 = makeAddr("user2");
23    weth.mint(user2, 50e18);
24    vm.startPrank(user2);
25    weth.approve(address(pool), 50e18);
26    pool.swapExactInput(
27        weth,
28        50 ether,
29        poolToken,
30        0,
31        uint64(block.timestamp)
32    );
33    vm.stopPrank();
34
35    // user expects to receive 5 WETH for expected amount of
36    // PoolTokens
37    weth.mint(user, 100e18); // mint more WETH to user to ensure
38    // they have enough
39    uint256 userStartingPoolTokenBalance = poolToken.balanceOf(user);
40    uint256 userStartingWethBalance = weth.balanceOf(user);
41    vm.startPrank(user);
42    weth.approve(address(pool), type(uint256).max);
43    pool.swapExactOutput(
44        weth,
45        poolToken,
46        outputAmount,
47        uint64(block.timestamp) + 1
48    );
49    vm.stopPrank();
50
51    uint256 userEndingPoolTokenBalance = poolToken.balanceOf(user);
52    uint256 userEndingWethBalance = weth.balanceOf(user);
53    vm.stopPrank();
54
55    uint256 userPoolTokensReceived = userEndingPoolTokenBalance -
56        userStartingPoolTokenBalance;
57    uint256 userWethSpent = userStartingWethBalance -
58        userEndingWethBalance;
```

```
53
54     console.log("Pool Tokens Received: ", userPoolTokensReceived);
55     console.log("Pool Tokens Expected: ", outputAmount);
56     console.log("WETH Expected to Send: ", expectedInputAmount);
57     console.log("WETH Spent: ", userWethSpent);
58
59     assert(userWethSpent > expectedInputAmount);
60
61 }
```

You will see that the user now has to spend more than double the amount of WETH to get the same amount of pool tokens.

Recommended Mitigation: Update `swapExactOutput` to include slippage protection.

```
1  function swapExactOutput(
2      IERC20 inputToken,
3      IERC20 outputToken,
4      uint256 outputAmount,
5 +     uint256 maxInputAmount
6      uint64 deadline
7  )
8      public
9      revertIfZero(outputAmount)
10     revertIfDeadlinePassed(deadline)
11     returns (uint256 inputAmount)
12 {
13     uint256 inputReserves = inputToken.balanceOf(address(this));
14     uint256 outputReserves = outputToken.balanceOf(address(this));
15
16     inputAmount = getInputAmountBasedOnOutput(
17         outputAmount,
18         inputReserves,
19         outputReserves
20     );
21
22 +     if(inputAmount > maxInputAmount) {
23 +         revert();
24 +     }
25
26     _swap(inputToken, inputAmount, outputToken, outputAmount);
27 }
```

[H-3] TSwapPool::sellPoolTokens mismatches input and output tokens causing users to receive the incorrect amount of tokens.

Description: The `sellPoolTokens` function is intended to allow users to easily sell pool tokens and receive WETH in exchange. Users indicate how many pool tokens they are willing to sell in the

'poolTokenAmount' parameter, but the function currently miscalculates the swapped amount.

This is because the `swapExactOutput` function is called whereas the `swapExactInput` function is the one that should be called since the users are specifying the exact amount of input pool tokens.

Impact: Users will swap the wrong amount of tokens, which is a severe disruption of the protocol functionality.

Proof of Code: You can add the following test to your test suite in `test/unit/TSwapPool.t.test`.

```
1 function testSellPoolTokensMishandlesSwapCalculation() public {
2     // LP deposit liquidity, 2 to 1 ratio weth to PT
3     vm.startPrank(liquidityProvider);
4     weth.approve(address(pool), 100e18);
5     poolToken.approve(address(pool), 100e18);
6     pool.deposit(100e18, 100e18, 50e18, uint64(block.timestamp));
7     vm.stopPrank();
8
9     // user attempts to swap 10 PT for WETH, should get around 20 WETH
10    uint256 startingBalanceWeth = weth.balanceOf(user);
11    uint256 startingBalancePoolToken = poolToken.balanceOf(user);
12
13    vm.startPrank(user);
14    uint256 poolTokensToSell = 10 ether;
15    poolToken.approve(address(pool), type(uint256).max);
16    pool.sellPoolTokens(poolTokensToSell);
17    vm.stopPrank();
18
19    uint256 endingBalanceWeth = weth.balanceOf(user);
20    uint256 endingBalancePoolToken = poolToken.balanceOf(user);
21    uint256 wethReceived = endingBalanceWeth - startingBalanceWeth;
22    uint256 poolTokensSpent = startingBalancePoolToken -
23        endingBalancePoolToken;
24
25    console.log("WETH Received:      ", wethReceived);
26    console.log("Pool Tokens Spent:   ", poolTokensSpent);
27    console.log("Pool Tokens To Sell: ", poolTokensToSell);
28
29    assert(poolTokensSpent != poolTokensToSell);
30 }
```

You will see that while the user intended to sell 10 pool tokens, he instead only sold around 5.5 pool tokens while receiving 10 weth instead.

Recommended Mitigation: Update `sellPoolTokens` such that it uses `swapExactInput` rather than `swapExactOutput`.

```
1 function sellPoolTokens(
```

```
2         uint256 poolTokenAmount
3 +         uint256 minWethAmount
4     ) external returns (uint256 wethAmount) {
5         return
6 -         swapExactOutput(
7 -             i_poolToken,
8 -             i_wethToken,
9 -             poolTokenAmount,
10 -            uint64(block.timestamp)
11 -        );
12 +        swapExactInput(
13 +            i_poolToken,
14 +            poolTokenAmount,
15 +            i_wethToken,
16 +            minWethAmount,
17 +            uint64(block.timestamp)
18 +        );
19     }
```

[H-4] In TSwapPool : : _swap, the extra sent to users after every 10 swaps breaks the protocol invariant of $x * y = k$.

Description: The protocol follows a strict invariant of $x * y = k$ Where:

- x : The balance of the pool Token
- y : The balance of WETH
- k : The constant product of the two balances.

This means that whenever the balances change in the protocol, the ration between the two amounts should remain constant, hence the k . However, this is broken when teh `_swap` function sends extra tokens every 10 swaps, meaning that over time, the protocol funds will be drained.

Impact: A user could maliciously drain the funds of the protocol by doing a lot of swaps and collecting the extra incentive given out by the protocol. The protocol's core invariant is broken.

Proof of Code: You can add the following test to your test suite in `test/unit/TSwapPool.t.test`.

```
1 function testSwapBreaksInvariant() public {
2     vm.startPrank(LiquidityProvider);
3     weth.approve(address(pool), 100e18);
4     poolToken.approve(address(pool), 100e18);
5     pool.deposit(100e18, 100e18, 100e18, uint64(block.timestamp));
6     vm.stopPrank();
7
8     uint256 outputWeth = 1e17;
9 }
```

```
10     vm.startPrank(user);
11     poolToken.approve(address(pool), type(uint256).max);
12     for(uint256 i = 0; i < 9; i++) {
13         pool.swapExactOutput(poolToken, weth, outputWeth, uint64(
14             block.timestamp));
15     }
16     int256 startingY = int256(weth.balanceOf(address(pool)));
17     int256 expectedDeltaY = int256(-1) * int256(outputWeth);
18     pool.swapExactOutput(poolToken, weth, outputWeth, uint64(block.
19         timestamp));
19     vm.stopPrank();
20
21     uint256 endingY = weth.balanceOf(address(pool));
22     int256 actualDeltaY = int256(endingY) - int256(startingY);
23     console.log("ExpectedDeltaY: ", expectedDeltaY);
24     console.log("ActualDeltaY:   ", actualDeltaY);
25     assert(expectedDeltaY != actualDeltaY);
26 }
```

This test will show, after the tenth swap, extra tokens will be sent to the user and the change in WETH tokens exceeds the expected change.

Recommended Mitigation: You should remove the logic in `_swap` that sends extra tokens every 10 swaps.

```
1 - uint256 private swap_count = 0;
2 - uint256 private constant SWAP_COUNT_MAX = 10;
3
4     ...
5
6     function _swap(
7         IERC20 inputToken,
8         uint256 inputAmount,
9         IERC20 outputToken,
10        uint256 outputAmount
11    ) private {
12        if (
13            _isUnknown(inputToken) ||
14            _isUnknown(outputToken) ||
15            inputToken == outputToken
16        ) {
17            revert TSwapPool__InvalidToken();
18        }
19
20        swap_count++;
21        if (swap_count >= SWAP_COUNT_MAX) {
22            swap_count = 0;
23            outputToken.safeTransfer(msg.sender, 1
24                _000_000_000_000_000_000);
25        }
26    }
```

```
25         emit Swap(  
26             msg.sender,  
27             inputToken,  
28             inputAmount,  
29             outputToken,  
30             outputAmount  
31         );  
32  
33         inputToken.safeTransferFrom(msg.sender, address(this),  
34             inputAmount);  
35         outputToken.safeTransfer(msg.sender, outputAmount);  
36     }
```

Medium

[M-1] TSwapPool::deposit is missing the deadline check, causing transactions to complete even after the deadline.

Description: The `deposit` function accepts a deadline parameter, which is the deadline for the transaction to be completed by in the documentation. This parameter is never used.

Impact: Transactions can be completed after the deadline that the function caller had set. Therefore, operations that add liquidity to the pool might be executed at unexpected times and in market conditions where the deposit rate is unfavorable. This also makes the transaction susceptible to MEV attacks.

Proof of Concept: The `deadline` parameter is unused.

Recommended Mitigation: Make the following change to the function within the function header.

```
1     function deposit(  
2         uint256 wethToDeposit,  
3         uint256 minimumLiquidityTokensToMint,  
4         uint256 maximumPoolTokensToDeposit,  
5         uint64 deadline  
6     )  
7     external  
8 +     revertIfDeadlinePasse(deadline)  
9     revertIfZero(wethToDeposit)  
10    returns (uint256 liquidityTokensToMint)  
11    { ... }
```

[M-2] Rebase, fee-on-transfer, and ERC777 tokens break protocol invariant.

Description: Rebase, fee-on-transfer, and ERC777 are weird ERC20s that will break the protocol invariant if they are deployed as the pool token in a TSwapPool.

Rebase tokens will update their value over time without transfers. A positive rebase will create a surplus of tokens and a negative rebase can make reserves lower than assumed, causing swaps to revert or set prices incorrectly.

Fee-on-transfer tokens will take some of the token balance when the token is transferred. The protocol will calculate the swap without accounting for the fee and break the protocol invariant.

ERC777 tokens leave the protocol open to reentrancies and should be avoided.

Impact: All of these weird ERC20 tokens will make your protocol vulnerable and break the core invariant of the protocol.

Proof of Code: This example will show how a fee-on-transfer token will break the protocol invariant. First, you will need a FeeOnTransferToken contract.

```
1 contract FeeOnTransferToken is ERC20 {
2     constructor() ERC20("Rebase", "RT") {}
3
4     function mint(address to, uint256 amount) external {
5         _mint(to, amount);
6     }
7
8     // Will take a 1e18 fee on every transfer
9     function _update(address from, address to, uint256 value) internal
10        virtual override {
11        // Skip fee on mint/burn (from==0 or to==0)
12        if (from != address(0) && to != address(0) && value > 0) {
13            super._update(from, address(0), 1e18);
14            super._update(from, to, value);
15        } else {
16            super._update(from, to, value);
17        }
18    }
```

Now, You can add the following test to your test suite in `test/unit/TSwapPool.t.test`.

```
1 function testFeeOnTransferTokenBreakInvariant() public {
2     FeeOnTransferToken feeToken = new FeeOnTransferToken();
3     pool = new TSwapPool(address(feeToken), address(weth), "Fee", "FT")
4     ;
5     feeToken.mint(liquidityProvider, 200e18);
6     feeToken.mint(user, 100e18);
```

```
7     weth.mint(user, 100e18);
8
9     vm.startPrank(liquidityProvider);
10    weth.approve(address(pool), 100e18);
11    feeToken.approve(address(pool), 101e18);
12    pool.deposit(100e18, 100e18, 100e18, uint64(block.timestamp));
13    vm.stopPrank();
14
15    vm.startPrank(user);
16    weth.approve(address(pool), type(uint256).max);
17    uint256 outputFeeToken = 1e18;
18    int256 startingX = int256(feeToken.balanceOf(address(pool)));
19    int256 expectedDeltaX = int256(-1) * int256(outputFeeToken);
20    pool.swapExactOutput( weth, feeToken, outputFeeToken, uint64(block.
        timestamp));
21    vm.stopPrank();
22
23    uint256 endingX = feeToken.balanceOf(address(pool));
24    int256 actualDeltaX = int256(endingX) - int256(startingX);
25    console.log("ExpectedDeltaX: ", expectedDeltaX);
26    console.log("ActualDeltaX:   ", actualDeltaX);
27    assert(expectedDeltaX != actualDeltaX);
28 }
```

Now you will see that the fee-on-transfer pool token will break the protocol invariant by transferring an extra token during the swap.

Recommended Mitigation: Do not create a pool with a rebase, fee-on-transfer, or ERC777 token.

Low

[L-1] TSwapPool::LiquidityAdded event in

TSwapPool::_addLiquidityMintAndTransfer has parameters out of order.

Description: When the `LiquidityAdded` event is emitted in `TSwapPool::_addLiquidityMintAndTransfer`, it logs value in an incorrect order. The `poolTokensToDeposit` value should go in the third parameter position, while the `wethToDeposit` value should go second.

Impact: Off chain functions may potentially malfunction do the incorrect event emission.

Recommended Mitigation: Update the parameters of the `LiquidityAdded` event emission in `TSwapPool::_addLiquidityMintAndTransfer`.

```
1 -   emit LiquidityAdded(msg.sender, poolTokensToDeposit, wethToDeposit)
    ;
2 +   emit LiquidityAdded(msg.sender,  wethToDeposit, poolTokensToDeposit
    );
```

[L-2] Default value returned by TSwapPool::swapExactInput results in an incorrect return value given.

Description: The `TSwapPool::swapExactInput` function is expected to return the actual amount of tokens bought by the caller. However, while it declares the named return value `output`, it is never assigned a value and there is not return statement used.

Impact: The return value will always be 0, giving incorrect information to the caller.

Proof of Code: You can add the following test to your test suite in `test/unit/TSwapPool.t.test`.

```
1 function testSwapExactInputWillAlwaysReturnZero() public {
2     // LP deposit liquidity
3     vm.startPrank(liquidityProvider);
4     weth.approve(address(pool), 100e18);
5     poolToken.approve(address(pool), 100e18);
6     pool.deposit(100e18, 100e18, 100e18, uint64(block.timestamp));
7     vm.stopPrank();
8
9     uint256 inputAmount = 1 ether;
10    uint256 inputReserves = poolToken.balanceOf(address(pool));
11    uint256 outputReserves = weth.balanceOf(address(pool));
12
13    uint256 expectedOutputAmount = pool.getOutputAmountBasedOnInput(
14        inputAmount,
15        inputReserves,
16        outputReserves
17    );
18
19    vm.startPrank(user);
20    poolToken.approve(address(pool), inputAmount);
21    uint256 swapReturnedOutputAmount = pool.swapExactInput(
22        poolToken,
23        inputAmount,
24        weth,
25        expectedOutputAmount,
26        uint64(block.timestamp)
27    );
28    vm.stopPrank();
29
30    console.log("expectedOutputAmount: ", expectedOutputAmount);
31    console.log("swapReturnedOutputAmount: ", swapReturnedOutputAmount)
32    ;
33    assert(swapReturnedOutputAmount == 0);
34 }
```

You will be able to see that the returned output from `swapExactInput` is 0.

Recommended Mitigation: You should either update the returned value `output` to `outputAmount` or change every instance of `outputAmount` in the function code to `output`.

```
1     function swapExactInput(  
2         IERC20 inputToken,    // e input tokens to swap / sell  
3         uint256 inputAmount,  // e amount of input tokens to swap /  
            sell  
4         IERC20 outputToken,   // e output tokens to receive / buy  
5         uint256 minOutputAmount, // e minimum amount of output tokens  
            to receive / buy  
6         uint64 deadline  
7     )  
8     public  
9     revertIfZero(inputAmount)  
10    revertIfDeadlinePassed(deadline)  
11    // @audit-low return not used, incorrect name  
12 -    returns (uint256 output)  
13 +    return (uint256 outputAmount)  
14    {  
15        ...  
16    }
```

Informational

[I-1] PoolFactory::PoolFactory__PoolDoesNotExist error is defined but never used and should be removed.

```
1 - error PoolFactory__PoolDoesNotExist(address tokenAddress);
```

[I-2] Missing zero address checks in the constructors of PoolFactory and TSwapPool when setting the token addresses.

You should make sure you are checking that the token addresses are not being set to the zero address.

PoolFactory:

```
1     constructor(address wethToken) {  
2 +         if(wethToken == address(0)) {  
3 +             revert();  
4 +         }  
5         i_wethToken = wethToken;  
6     }
```


TSwapPool:

```
1     constructor(  
2         address poolToken,  
3         address wethToken,  
4         string memory liquidityTokenName,  
5         string memory liquidityTokenSymbol  
6     ) ERC20(liquidityTokenName, liquidityTokenSymbol) {  
7 +     if(poolToken == address(0) || wethToken == address(0)) {  
8 +         revert();  
9 +     }  
10     i_wethToken = IERC20(wethToken);  
11     i_poolToken = IERC20(poolToken);  
12 }
```

[I-3] PoolFactory::liquidityTokenSymbol should use .symbol() instead of .name() when creating the symbol for the liquidity provider token.

```
1 - string memory liquidityTokenSymbol = string.concat("ts", IERC20(  
    tokenAddress).name());  
2 + string memory liquidityTokenSymbol = string.concat("ts", IERC20(  
    tokenAddress).symbol());
```

[I-4] TSwapPool::deposit updates internal state after an external call, which does not follow CEI.

Move the internal state change of `liquidityTokensToMint`, which is the return value, before the external calls in `_addLiquidityMintAndTransfer` to follow best practices.

```
1 + liquidityTokensToMint = wethToDeposit;  
2     _addLiquidityMintAndTransfer(  
3         wethToDeposit,  
4         maximumPoolTokensToDeposit,  
5         wethToDeposit  
6     );  
7 - liquidityTokensToMint = wethToDeposit;
```

[I-5] “Magic” numbers in TSwapPool should be set as constants.

`TSwapPool::getOutputAmountBasedOnInput` and `TSwapPool::getInputAmountBasedOnOutput` both contain literals that should be instead defined as constants.

For example, in `TSwapPool::getOutputAmountBasedOnInput`, you have:

```
1 uint256 inputAmountMinusFee = inputAmount * 997;  
2 uint256 numerator = inputAmountMinusFee * outputReserves;  
3 uint256 denominator = (inputReserves * 1000) + inputAmountMinusFee;  
4 return numerator / denominator;
```

Instead, these values should be defined as constants at the beginning of the contract:

```
1 uint256 public constant FEE_ADJUSTMENT= 997;  
2 uint256 public constant FEE_PRECISION = 1000;  
3  
4 ...  
5  
6 uint256 inputAmountMinusFee = inputAmount * FEE_ADJUSTMENT;  
7 uint256 numerator = inputAmountMinusFee * outputReserves;  
8 uint256 denominator = (inputReserves * FEE_PRECISION) +  
    inputAmountMinusFee;  
9 return numerator / denominator;
```

This should be done for other instances of “magic” numbers in the contract.

[I-6] TSwapPool::swapExactInput does not have any natspec.

You should have natspec explaining what the `TSwapPool::swapExactInput` function does and what the input parameters are as one of the main contract functions used by external entities.

[I-7] TSwapPool::swapExactOutput natspec misses the deadline parameter.

You should add natspec for the `TSwapPool::swapExactOutput` function to explain the deadline parameter.

Gas

[G-1] TSwapPool::MINIMUM_WETH_LIQUIDITY is a constant and should not be emitted within a revert.

To save some gas on reverts, `TSwapPool::TSwapPool__WethDepositAmountTooLow` should only contain the parameter for the WETH amount that was too low to deposit.

```
1 error TSwapPool__WethDepositAmountTooLow(  
2 -     uint256 minimumWethDeposit,  
3     uint256 wethToDeposit  
4 );
```

```
5
6     ...
7
8     revert TSwapPool__WethDepositAmountTooLow(
9 -         MINIMUM_WETH_LIQUIDITY,
10         wethToDeposit
11     );
```

[G-2] TSwapPool::deposit contains an unused variable, which should be removed to save gas.

```
1 - uint256 poolTokenReserves = i_poolToken.balanceOf(address(this));
```

[G-3] TSwapPool::swapExactInput should be marked as external.

`TSwapPool::swapExactInput` is never called internally, so you can save gas by marking as external.